/** FC合集,把基础按键都放到不用的地方,所有使用的键都是热键,这样才满足
  多个游戏的需求.基础按键依次为:a,b,start,select
 
  By setycyas @2025-05-14
*/

/* 全局设定
*/ {
#Warn
#MaxThreadsPerHotkey 1
SetWorkingDir(A_ScriptDir)
#Include "..\Class\FaceDetector.2.ahk"
#Include "..\Class\CommonMacro.2.ahk"
#Include "..\Class\CommonUI.2.ahk"
#Include "..\Class\MyFuns.2.ahk"
#Include "..\Class\MyEmu.2.ahk"
#Include "..\Class\OrderMap.2.ahk"
;#Include "..\Class\DrawWindow.2.ahk"
;SendMode("Event")
SendMode("Input")
}

/* 制作Fc的宏类
*/ class FCMacro extends CommonMacro {

  pressDuration := 30

  /* 函数
  */

  /* 新增的宏函数
  */
  ;单发
  _单宏模拟(i, key){
    local vk := GetKeyVK(MyFuns.getHotkey(key))
    this._down(i)
	  sleep(this.autoFireDelay)
    loop {
	    sleep(this.autoFireDelay)
    } until (not MyFuns.getVkState(vk))
    this._up(i)
  }
  _单发A(key){
    this._单宏模拟(1, key)
  }
  _单发B(key){
    this._单宏模拟(2, key)
  }
  _单发start(key){
    this._单宏模拟(3, key)
  }
  ;ab
  _AB(key){
    this._down(1)
    this._down(2)
    sleep(60)
    this._up(1)
    this._up(2)
  }
  ;蓄力
  _蓄力A(key){
    this._down(1)
  }
  _蓄力B(key){
    this._down(2)
  }
  ;轻微位移用
  _右微移(key){
    MyFuns.press("right", 50)
    sleep(50)
  }
  _左微移(key){
    MyFuns.press("left", 50)
    sleep(50)
  }
  ;其他
  _后跳(key){
    local back := this._face.back()
    ;MyFuns.pressKeys([this._keys[1], back], [this.testParams[1], 50])
    this._down(1)
    MyFuns.keyDown(back)
    sleep(50)
    this._up(1)
    MyFuns.keyUp(back)
  }
  _上B(key){
    MyFuns.keyDown("up")
    this._press(2)
    MyFuns.keyUp("up")	  
  }
  ;成龙之龙用
  _小AB(key){
  	this._down(1)
  	sleep(30)
  	this._down(2)
  	sleep(30)
  	this._up(1)
  	this._up(2)
  }
  ;SD快打旋风专用
  ;连发ab
  _按住B后A连发(key){
  	MyFuns.clearDirections()
    this._down(2)
    sleep(30)
    this._连发A(key)
    sleep(30)
    this._up(2)
  }
  ;B+前,必杀技
  _B前(key){
  	local face := this._face.face()
    this._down(2)
    sleep(30)
    MyFuns.press(face, 50)
    this._up(2)	
  }
  ;Guy的后跳必杀,按住触发放手解除
  _Guy后跳必杀(key){
  	;local face := this._face.face()
    local back := this._face.back()	
    this._press(1)
    sleep(30)
    MyFuns.press(back, 50)
    sleep(30)
    MyFuns.keyDown("up")
    sleep(50)
    this._连发B(key)
    MyFuns.keyUp("up")
  }
  ;cody升龙拳
  _Cody升龙拳(key){
    this._B前(key)
    sleep(this.testParams[1])
    MyFuns.keyDown("up")
    sleep(30)
    this._连发B(key)
    MyFuns.keyUp("up")	
  }
  
  /* 重载宏接口
  */
  _setMacro(){
    static obm := ObjBindMethod ;函数名简写,制作对象绑定函数
    local m := OrderMap() ;this.macroMap的简写
    ;制作宏相关Map
    this.macroMap := m
    ;通用
    m["连发A"] := obm(this, "_连发A")
    m["连发B"] := obm(this, "_连发B")
    m["单发A"] := obm(this, "_单发A")
    m["单发B"] := obm(this, "_单发B")
    m["单发start"] := obm(this, "_单发start")
    m["蓄力A"] := obm(this, "_蓄力A")
    m["蓄力B"] := obm(this, "_蓄力B")
    m["AB"] := obm(this, "_AB")
    m["后跳"] := obm(this, "_后跳")
    m["左微移"] := obm(this, "_左微移")
    m["右微移"] := obm(this, "_右微移")
    m["空函数"] := obm(this, "_空函数")
    ;上+攻击键出招类通用
    m["上B"] := obm(this, "_上B")
    ;成龙之龙用
    m["小AB"] := obm(this, "_小AB")
    ;SD快打旋风用
    m["按住B后A连发"] := obm(this, "_按住B后A连发")
    m["B前"] := obm(this, "_B前")
    m["Guy后跳必杀"] := obm(this, "_Guy后跳必杀")
    m["Cody升龙拳"] := obm(this, "_Cody升龙拳")
    ;制作指定宏顺序的数组
    this.macroArr := m.getKeys()
  }
}

/* 制作FC专用UI
*/
class FCUI extends CommonUI {

  HOTKEY_NUM := 8
  EXE_ARRAY := ["fceux.exe", MyEmu.FBNEO64, MyEmu.MAME, MyEmu.XZONE]
  
  /* 构造函数
  */
  __New(macro){
    super.__New("FC合集", macro)
    local preset := OrderMap() ;部分预设表
    this._preset := OrderMap()
    this._macro.UI := this ;绑定UI
    ;设置preset和提升蓄力宏的优先级
    ;部分预设
    preset["B攻连A跳单"] := ["连发B", "单发A", "单发B", "连发A"]
    preset["A攻连B跳单"] := ["连发A", "单发B", "单发A", "连发B"]
    preset["后4按键标准"] := ["空函数", "左微移", "右微移", "单发start"] ;a,w,e,space按键对应
    ;实际预设
    this._preset["B攻连A跳单"] := MyFuns.arrCombine(preset["B攻连A跳单"], preset["后4按键标准"])
    this._preset["A攻连B跳单"] := MyFuns.arrCombine(preset["A攻连B跳单"], preset["后4按键标准"])
    this._preset["上B"] := this._preset["B攻连A跳单"].Clone()
    this._preset["上B"][5] := "上B"
    this._preset["三目童子"] := this._preset["B攻连A跳单"].Clone()
    this._preset["三目童子"][3] := "蓄力B"
    this._preset["三目童子"][5] := "单发B"	
    this._preset["成龙之龙"] := this._preset["B攻连A跳单"].Clone()
    this._preset["成龙之龙"][3] := "蓄力B"	
    this._preset["成龙之龙"][5] := "小AB"
    this._preset["热血物语"] := ["连发A", "AB", "连发B", "单发A", 
      "连发B", "左微移", "右微移", "单发start"
    ]
    this._preset["帝国战机"] := this._preset["B攻连A跳单"].Clone()
    this._preset["帝国战机"][8] := "AB"
    this._preset["SD快打旋风"] := MyFuns.arrCombine(preset["B攻连A跳单"],
      ["B前", "Cody升龙拳", "Guy后跳必杀", "按住B后A连发"])
  }

  /* 重写接口
  */
  ;自定义gui组件标签页,增加一个挡板召唤键
  _customGui(){
    super._customGui()
  }

  /* 重写setup,更改可选的exe,读入预设
  */
  setup(){
    super.setup()
    this._myGui.opt("+AlwaysOnTop")
    this._myGui.show("x1100 y230")
    this.readPresetMenu("B攻连A跳单")
  }
  
  /* 自定义函数
  */
  
}

/* 运行当前脚本
*/ 
if (A_ScriptName == "FC合集.2.ahk") {
  ;iconFile := "..\Icon\怒首领蜂最大往生_光头像01.jpg"
  ;TraySetIcon(iconFile, , 1)

  face := FaceDetector() ;方向检测对象
  macro := FCMacro(face, ["i","o","space","f"]) ;宏对象,有界面时第二个参数可以随便写但长度不要小于2
  ;新建界面对象并显示
  ui := FCUI(macro)
  ui.setup()
  ;更新组件内容
  ui._exeComboBox.Choose("fceux.exe")
  ;ui._exeComboBox.Choose(MyEmu.FBNEO64)
  ui._gameKeyEdit.Text := "i,o,r,f"
  ui._hotkeyEdit.Text := "~s,~d,~x,~c,~a,~w,~e,~space"
  ui._fpsEdit.Text := 15
  ui._testEdit.Text := "50,100,30,30"
  ui.readPresetMenu("B攻连A跳单")
}
